



Let's put a bunch of components together to create an automatic light system: the darker it gets, the brighter an LED becomes.

Before it will work accurately, we'll need to *calibrate* it for its particular location. We use the button to help perform the calibration.

In the code we'll use two new variable types:

- **float**: a number which includes decimal places (unlike an **int** which can only be a whole number)
- **bool**: can only be true or false

We'll also use a new loop: the **while** loop

```
ldr
#define LDR A2 // select the input pin for ldr
#define LED 6 //LED
#define BUTTON 8 //Button
int sensorValue = 0; // variable to store the value coming from the sensor
bool need_calib = true; // used to trigger calibration at start
int v_min; // volatge when LDR is covered
int v_max; // voltage when LDR is in normal light
float v_factor; // scaling factor for PWM of LED
float v_range; // range of voltage between light/dark
void setup() {
  pinMode(LDR, INPUT);
  pinMode(LED, OUTPUT);
  pinMode(BUTTON, INPUT);
  Serial.begin(9600); //sets serial port for communication
}

void loop() {
  if (need_calib) {
    Serial.println("Put LDR in a bright place and then press the button for 1 second");
    while (digitalRead(BUTTON) == LOW) { //wait for button to be pressed
      delay(100);
    }
    v_max = analogRead(LDR);
    Serial.print("Voltage in bright light is: ");
    Serial.println(v_max);
    while (digitalRead(BUTTON) == HIGH) { //wait for button to be released
      delay(100);
    }

    delay(1000);
    Serial.println("Thanks. Now cover the LDR and press the button again.");
    while (digitalRead(BUTTON) == LOW) { //wait for button to be pressed
      delay(10);
    }
    v_min = analogRead(LDR);
    Serial.print("Voltage in darkness is: ");
    Serial.println(v_min);
    v_range = v_min - v_max; //calculate the range of voltages
    v_factor = 256/v_range; // remember the max PWM value is 255
    Serial.print("Calculated factor as: ");
    Serial.println(v_factor);
    Serial.println("Calibration complete, Bye!");
    need_calib = false; // now we've calibrated, set this so we don't do it again
  }

  sensorValue = analogRead(LDR); // read the value from the sensor
  Serial.println(sensorValue); //prints the values coming from the sensor on the screen
  delay(100);
  // set LED brightness depending on how dark LDR says it is
  analogWrite(LED, abs((sensorValue-v_max)*v_factor));
}
```